
Zadání úlohy:

Úloha 1	 Strukturované programování, složitější algoritmy, dynamické proměnné

Napište program (programový systém), který bude řešit úlohu, která má číslo shodné s poslední číslicí vašeho rodného čísla. Program realizujte v Pascalu.

Zadání 2



a) Založte binární soubor s následující strukturou věty:

číslo skladu		číslo 1 až 10

číslo materiálu		pětimístné číslo

kód pohybu		P - příjem, V - výdej

množství		celé číslo





b) Setřiďte jej podle čísla materiálu s použitím indexů a binárního stromu. Vytvořte fyzicky nový soubor a vypište jej na obrazovku



c) Algoritmem typu skupinové operace zpracujte setříděný soubor tak, že výstupem je textový soubor se sumarizovanými pohyby za určitý materiál



	číslo materiálu   suma množství (+-)

Popis řešení úlohy:

Celou úlohu jsem rozdělil na tři nezávislé programy, kde každý program řeší jeden bod zadání. Lze tak snadněji sledovat a testovat funkci jednotlivých částí programu. Protože však všechny části programu používají podobné datové struktury, lze jednoduše všechny tři části spojit v celek.

První program VYTVOŘ.PAS zakládá binární soubor požadované struktury. Pro zjednodušení a urychlení program využívá generátoru náhodných čísel: generuje jednotlivé položky věty, které vkládá do proměnné SKLAD typu record (viz. Datové struktury typ TMat), kterou následně uloží jako větu binárního souboru MATERIAL.BIN. Generování se opakuje podle předem zvoleného počtu iterací v cyklu FOR. Program zároveň provádí současný výpis generovaných hodnot na obrazovku (vhodné například ke kontrole při trasování). Po uložení požadovaného počtu vět program uzavře výstupní soubor a ukončí svou činnost.

Druhou a největší částí je program TŘÍDĚNÍ.PAS, který provádí vlastní setřídění souboru MATERIAL.BIN pomocí indexů a binárního stromu. Program načítá jednotlivé věty ze vstupního souboru do proměnné SKLAD a podle položky SKLAD.cMat vytváří dynamicky binární strom. Ten je tvořen tak, že v levém podstromu jsou prvky s hodnotou klíče menší, než rodič a v pravém podstromu prvky s hodnotou klíče větší nebo rovnou hodnotě klíče rodiče. Struktura jednotl. prvku binárního stromu obsahuje také pozici záznamu s odpovídajícím klíčem ve zdrojovém souboru. Po vytvoření celého binárního stromu se volá rekurentní procedura SETRID, která postupně prochází binárním stromem metodou INORDER. Přitom z každého prvku zjistí pozici požadované věty ve vstupním souboru, nastaví na ni pozici (SEEK), přečte ji a uloží do výstupního souboru MAT_TRID.BIN. Průchodem celého binárního stromu metodou INORDER tak získáme setříděný soubor podle klíče, jímž je zde číslo materiálu (položka cMat). Nakonec program v přehledné formě vypíše setříděný soubor MAT_TRID.BIN na obrazovku.

Třetí program VÝTISK.PAS zpracovává setříděný binární soubor tak, že výsledkem je textový soubor se zadanou strukturou věty: t.j. číslo materiálu a suma množství (vlastně toková veličina, která vznikne odečtením souhrnu výdajů od souhrnu příjmů za jednotlivé materiály). Program používá skupinové operace: Program sleduje položku cMat načítané věty a dokud je tato stejná, přičítá resp. odečítá (podle položky kod téže věty) hodnotu mnoz, ve které je uloženo množství, k proměnné Celkem. Změní-li se položka cMat, program vypíše do výstupního textového souboru VYTISK.TXT číslo zpracovávaného materiálu a celkovou změnu množství, kterou následně nuluje.

Všechny tři programy jsou připraveny ke spojení v jeden komplexní celek, proto část zpracovávající výstup části předchozí nekontroluje existenci vstupního souboru a nastavení jeho atributů (předpokládá, ze soubor požadovaného jména v aktuálním adresáři existuje a lze z něj číst, resp. do něj zapisovat). Programy také pro zjednodušení předpokládají, že na cílovém disku je dostatek místa a že na něj lze zapisovat.

Datové struktury:

Struktura binárních souborů MATERIAL.BIN a MAT_TRID.BIN je následující:

	číslo skladu		číslo 1 až 10

	číslo materiálu	pětimístné číslo

	kód pohybu		P - příjem, V - výdej

	množství		celé číslo

Textový soubor VYTISK.TXT má strukturu:

	číslo materiálu   suma množství



Všechny části používají v určité formě strukturu věty TMat, která je definována takto:

type pohyb=set of (P,V);	množina definující možné kódy pohybu

     Tmat=record

      cSkladu:1..10;	číslo skladu

      cMat:1..99999;	číslo materiálu

      kod:pohyb;	kód pohybu

      mnoz:byte;	množství

     end;

Program TRIDENI.PAS navíc pro založení binárního stromu definuje typ Strom:

Strom=^Tree;	Strom je tvořen prvky stromu

Tree=record	struktura prvku stromu:

     cMat,pozice:word;	klíč (číslo materiálu), pozice v souboru

     Levy,Pravy:Strom;	ukazatelé na potomky prvku

    end;

Program VYTVOR.PAS deklaruje proměnné takto:

Var a,i,j:Byte;	pomocné proměnné a indexy

    b:word;	pomocná proměnná pro cyklus FOR

    sklad:TMat;	proměnná ukládaná do bin. souboru

    f:file of Tmat;	výstupní soubor

Proměnné programu TRIDENI.PAS jsou:

var ROOT:STROM;	proměnná pro binární strom

    sklad:TMat;	vstupně-výstupní prom.

    f,ff:file of Tmat;	vstupní a výstupní binární soubory

    i,j:Byte;	pomocné proměnné a indexy

    ch:Char;	pomocná proměnná pro čtení z klávesnice

Program VYTISK.PAS používá tyto proměnné:

Var Celkem:Integer;	suma

    old,sklad:TMat;	vstupně-výstupní prom.

    f:file of Tmat;	vstupní binární soubor

    ff:text;	výstupní textový soubor

Grafická dokumentace:



Zdrojový program:

První program vytváří binární soubor s požadovanou strukturou věty.

VYTVOR.PAS��Program VYTVOR; {Vytvori soubor MATERIAL.BIN}



Uses Crt;



Const Copyright=' Copyright (c) 1999, Michal Kec';

	fn='Material.bin';

	MAX=100;



type pohyb=set of (P,V);

     TMat=record

      cSkladu:1..10;

      cMat:1..99999;

      kod:pohyb;

      mnoz:byte;

     end;



Var a,i,j:Byte;

    b:word;

    h:Char;

    sklad:TMat;

    f:file of TMat;





Begin

 HighVideo;				{Nastaveni parametru obrazovky}

 TextColor(yellow);

 TextBackground(blue);

 ClrScr;

 WriteLn;

 WriteLn (Copyright);

 window(1,3,80,25);

 textbackground(black);

 clrscr;

 TextColor(green);

 randomize;				{Inicializace generatoru nahodnych cisel}

 assign(f,fn);

 rewrite(f);				{Zalozeni vystupniho souboru}

 j:=0;

 for i:=1 to 80 do write('-');

 for B:=1 to MAX do			{Iterace}

  begin;

   if j=4 then

    begin

     j:=0;

     clrscr;

     for i:=1 to 80 do write('-');

    end;

   inc(j);

   with sklad do			{Generovani nahodnych hodnot}

    begin				{Zaroven vypisuje na obrazovku}

     write(' Zadej cislo skladu [1..10]        ');

     cSkladu:=(random(10)+1); writeln(cskladu);

     write(' Zadej cislo materialu [1..99999]  ');

     cmat:=(random(MAX div 3)*1108+10012); writeln(cmat);

     write(' Zadej kod pohybu [p,v]            ');

     a:=random(2);

     if a=0 then begin kod:=[v]; writeln('Vydej'); end;

     if a=1 then begin kod:=[p]; writeln('Prijem'); end;

     write(' Zadej mnozstvi [1..255]           ');

     mnoz:=random(245)+10; writeln(mnoz);

     for i:=1 to 80 do write('-');	{Oddeli zaznam}

    end;

   write(f,sklad);			{Zapsani vety do souboru}

  end;

 close(f);				{Uzavreni souboru}

 TextColor(yellow);

 write (' ',B,' zaznamu...');

End.

{Created by MiK}

��

Druhý program třídí binární soubor.

TRIDENI.PAS��program TRIDENI; {Setridi MATERIAL.BIN pomoci indexu a bin. stromu}

		 {Vysledek ulozi do souboru MAT_TRID.BIN}

uses Crt;



Const Copyright=' Copyright (c) 1999, Michal Kec';

	fn='Material.bin';

	fs='Mat_trid.bin';



type pohyb=set of (P,V);

     TMat=record

      cSkladu:1..10;

      cMat:1..99999;

      kod:pohyb;

      mnoz:byte;

     end;



   Strom=^Tree;

   Tree=record

     cMat,pozice:word;

     Levy,Pravy:Strom;

    end; 



var Root,Rodic:Strom;

    Sklad:TMat;

    f,ff:file of TMat;

    i,j:Byte;

    ch:Char;





procedure Novy(sklad:TMat; var Rodic:Strom);	{pripoji novy prvek ke stromu}

 begin

  new(Rodic);				{vytvoreni noveho prvku}

  with Rodic^ do

   begin

    cMat:=sklad.cMat;			{naplneni klice}

    pozice:=filepos(f);			{naplneni pozice v souboru}

    Levy:=nil;				{prvek nema zadne deti}

    Pravy:=nil

   end

 end;



procedure Umisti(sklad:TMat; var Rodic:STROM);	{umisti novy prvek na spravne misto stromu}

 begin

  if Rodic=nil then Novy(sklad,Rodic)     {jestlize rodic nema deti,pripoj novy}

   else

   if sklad.cmat<Rodic^.cMat then Umisti(sklad,Rodic^.Levy)	{vloz doleva}

    else Umisti(sklad,Rodic^.Pravy)				{vloz doprava}

 end;



procedure Setrid(Root:STROM);		{prochazi stromem metodou Inorder}

 begin

  if Root<>nil then			{prochazi az na listy}

   begin

    Setrid(Root^.Levy);			{nejprve levy podstrom}

    seek(f,Root^.pozice-1);		{nastavi pozici na hodnotu v seznamu}

    read(f,sklad);			{nacte vetu ze zdroje}

    write(ff,sklad);			{a ulozi ji do vystupu}

    Setrid(Root^.Pravy)			{prochazi pravou vetev}

   end

 end;





Begin					{Zacatek tela programu}

 HighVideo;				{Nastaveni parametru obrazovky}

 TextColor(yellow);

 TextBackground(blue);

 ClrScr;

 WriteLn;

 Writeln(copyright);

 write (' Cekejte prosim, tridim...');

 assign(f,fn);

 reset(f);				{Otevreni vstupniho souboru}

 if not eof(f) then read(f,sklad);	{Nacitani vety}

 Novy(sklad,Root);			{Zalozeni korene stromu }

 while not eof(f) do

  begin

   write('.');

   read(f,sklad);			{Nacteni vety}

   Rodic:=Root;

   Umisti(sklad,Rodic)			{Umisti nactenou vetu do stromu}

  end;

 assign(ff,fs);

 rewrite(ff);				{Zalozeni vystupniho souboru}

 setrid(Root);				{Prochazi strom inorder a tridi soubor}

 close(f);				{Uzavreni vstupniho souboru}

 close(ff);				{Uzavreni vystupniho souboru}

 writeln;

 Write(' Hotovo. Nyni vypisu setrideny soubor');

 delay (3000);



 reset(ff);				{Otevreni setrideneho souboru}

 gotoxy(69,2);				{Nastaveni atributu obrazovky}

 write(filesize(ff),' zaznamu');

 window(1,3,80,25);

 textbackground(black);

 TextColor(green);

 clrscr;

 for i:=1 to 80 do write('-');

 j:=0;

 ch:=#0;

 while not eof(ff) do			{Opakuje do konce souboru}

  begin

   read(ff,sklad);			{Nacte vetu}

   if j=4 then				{Jednou za ctyri vety pocka a smaze obr.}

    begin

     j:=0;

     gotoxy(2,23);

     TextColor(lightgray); Write('Esc=ukonci, Space=pokracuj'); TextColor(green);

     gotoxy(70,23);

     write(filepos(ff)div 4 ,'/',filesize(ff) div 4);

     ch:=readkey;

     if ch=#27 then exit;		{Byl-li zmacknut Esc, skonci}

     clrscr;

     for i:=1 to 80 do write('-');

    end;

   inc(j);

   with sklad do			{Vypise obsah vety na obrazovku}

    begin

     write (' Cislo skladu: ');    TextColor(yellow); writeln(cSkladu); TextColor(green);

     write (' Cislo materialu: '); TextColor(yellow); writeln(cmat);    TextColor(green);

     write (' Kod pohybu: ');      TextColor(yellow);

     if kod=[v] then writeln('Vydej ');

     if kod=[p] then writeln('Prijem');

     TextColor(green);

     write(' Mnozstvi: ');         TextColor(yellow); writeln(mnoz);    TextColor(green);

     for i:=1 to 80 do write('-');

    end;

   end;

  close(ff);

 writeln('      >>>> Tadydadyda. To je vse pratele! <<<<');

 TextColor(lightgray); write ('Enter...');

 readln;

End.

{Created by MiK}

��

Třetí program generuje výsledný textový soubor.

VYTISK.PAS��Program VYTISK; {Prevede bin. soub. na textovy se zadanou strukturou vety}



Uses Crt;



Const Copyright=' Copyright (c) 1999, Michal Kec';

	fs='Mat_trid.bin';

	ft='Vytisk.txt';



type pohyb=set of (P,V);

     TMat=record

      cSkladu:1..10;

      cMat:1..99999;

      kod:pohyb;

      mnoz:byte;

     end;



Var Celkem:Integer;

    old,sklad:TMat;

    f:file of TMat;

    ff:text;





Begin

 HighVideo;				{Nastaveni parametru obrazovky}

 TextColor(yellow);

 TextBackground(blue);

 ClrScr;

 WriteLn;

 WriteLn (Copyright);

 writeln(' Zpracovavam binarni soubor "',fs,'".');

 assign(f,fs);

 reset(f);				{Otevreni vstupniho souboru}

 assign(ff,ft);

 rewrite(ff);				{Otevreni vystupniho souboru}

 writeln (ff,'Cislo materialu   suma mnozstvi');	{Tisk zahlavi}

 Celkem:=0;

 while not eof(f) do			{Provadi do konce souboru}

  begin

   read(f,sklad);			{Nacteni vety}

   with sklad do

    begin

     if (old.cmat<>cmat) and (Celkem<>0) then	{Pri zmene cMat}

      begin

       writeln(ff,'   ',old.cmat:5,'              ',Celkem:5);	{Vypise mnozstvi}

       Celkem:=0;			{Nulovani promene}

      end;

     if kod=[V] then Celkem:=Celkem-mnoz;	{Spocte celkove mnozstvi}

     if kod=[P] then Celkem:=Celkem+mnoz;

     old:=sklad;			{Ulozeni predchoziho}

    end;

  end;

 close(f);				{Uzavreni souboru}

 close(ff);

 writeln(' Vysledny soubor "',ft,'" ulozen.');

End.

{Created by MiK}

��


